How to prepare interactive Mathematica demonstrations for classroom

János Karsai

University of Szeged

http://silver.szote.u-szeged.hu/~karsai
karsai@dmi.u-szeged.hu

Some examples from the Wolfram demonstration project

What I like

Logistic Equation	(source)
Partial Derivatives in 3 *D*	(source)
Directional Derivatives in 3 *D*	(source)
Slope Fields	(source)
Phase Plane Plot of the Van der Pol Differential Equation	(source)
Phase Portrait and Field Directions of 2 *D* Linear Systems of ODEs	(source)
Visualizing the Gradient Vector	(source)

◄ | ►

They are OK, but...

Two – Dimensional Linear Systems	(source)
Linear Transformation with Given Eigenvectors	(source)
Driven Damped Oscillator	(source)
Double Integral for Volume	(source)
Constrained Optimization	(source)
Bifurcations in First – Order ODEs	(source)
Brownian Motion in 2 *D* and the Fokker – Planck Equation	(source)

◄ | ►

I do not like for some reason

3 *D* Vector Fields	(source)
Cauchy – Schwarz Inequality for Integrals	(source)
From Vector to Plane	(source)
Dynamic Behavior of *a* Simple Canonical System	(source)
Saddle Points and Inflection Points	(source)

◄ | ►

Some own more complex developments

Interactive Curve Fitting	(source)
Competition for Territory : The Levins Model for Two Species	(source)
Virtual Flowers	(source)
1 D ODE explorer	(source)
Intravascular Dosing, Version 1	(source)
Intravascular Dosing, Realistic Version	(source)
Extravascular Dosing	(source)

◄ | ►

Start out of "one-minute" demonstrations

◄ | ►

Example: Plot power functions

positive integer exponents
negative powers
rational powers
general real powers
Illustrate inverse, reciprocal...
... or....
study the local-global behavior
... or...

◻ *Simplest version*

```
Manipulate[Plot[x^n, {x, -1.5, 1.5},
  AspectRatio → Automatic, PlotRange → {-2.1, 2.1}], {n, 0, 20, 1}]
```



◻ A stroboscopic plot of powers

```
Manipulate[Plot[Map[x^# &, Range[n]], {x, 0, 1.5},
  AspectRatio → Automatic, PlotRange → {0, 1.5}], {n, 1, 20, 1}]
```



◻ A power and its inverse

```
Manipulate[Plot[{x^n, x^(1/n)}, {x, 0, 1.5},
  AspectRatio → Automatic, PlotRange → {0, 1.5},
  PlotLabel -> TraditionalForm[{x^n, x^(1/n)}]], {n, 1, 10, 1}]
```



▫ Use Setterbar

```
Manipulate[Plot[x^n, {x, -1.5, 1.5}, AspectRatio → Automatic,
  PlotRange → {-2.1, 2.1}, PlotLabel -> TraditionalForm[x^n]],
 {n, Range[10], SetterBar}]
```



▫ *Use Togglerbar*

```
Manipulate[
 Plot[Map[x^# &, n], {x, -1.5, 1.5}, AspectRatio → Automatic,
  PlotRange → {-2.1, 2.1}, PlotLabel -> TraditionalForm[x^n]],
 {{n, {}}, Range[10], TogglerBar}]
```



Example: Simple transformations over functions

Apply simple transformations on functions:
Stretch, compress, mirror: $a\, f(x)$, $f(c\, x)$
Translate the graph by the vector $\{p, q\}$: $f(x-p)+q$
...or find the formula for the transformed graph.

Consider the translation

A simple version

```
Manipulate[
 Plot[{x^2, (x - p)^2 + q}, {x, -r, r},
  AspectRatio → Automatic, PlotRange → {{-r, r}, {-r, r}}],
 "Translation of the graph of x² by the vector {p,q}",
 {{p, 0}, -2, 2}, {{q, 0}, -2, 2},
 Delimiter, {{r, 1, "Range"}, 1, 10}, ControlPlacement → Left]
```



◻ *A little bit better with the transformed axes*

```
Manipulate[
 Plot[{x^2, (x - p)^2 + q}, {x, -r, r},
  AspectRatio → Automatic, PlotRange → {{-r, r}, {-r, r}},
  Prolog → {Opacity[0.5], Arrow[{{0, 0}, {p, q}}],
    Line[{{{-r, q}, {r, q}}, {{p, -r}, {p, r}}}]}],
 "Translation of the graph of x² by the vector {p,q}",
 {{p, 0}, -2, 2}, {{q, 0}, -2, 2},
 Delimiter, {{r, 1, "Range"}, 1, 10}]
```



Use 2D slider

```
Manipulate[Plot[{x^2, (x - pq[[1]])^2 + pq[[2]]}, {x, -r, r},
  AspectRatio → Automatic, PlotRange → {{-r, r}, {-r, r}},
  Prolog → {Opacity[0.5], Arrow[{{0, 0}, pq}], Line[
     {{{-r, pq[[2]]}, {r, pq[[2]]}}, {{pq[[1]], -r}, {pq[[1]], r}}}]}],
 "Translation of the graph of x^2 by the vector {p,q}",
 {{pq, {0, 0}}, {-2, -2}, {2, 2}},
 Delimiter, {{r, 1, "Range"}, 1, 10}]
```


Translation of the graph of x^2 by the vector {p,q}
pq
Range
2
1
−1
−2
−2
−1
1
2

▫ *Use 2D Locator and type the formula*

```
Manipulate[
 Column[{{x^2, (x - pq[[1]])^2 + pq[[2]]},
   Plot[{x^2, (x - pq[[1]])^2 + pq[[2]]}, {x, -r, r},
    AspectRatio → Automatic,
    PlotRange → {{-r, r}, {-r, r}}, Prolog → {Opacity[0.5],
      Arrow[{{0, 0}, pq}], Line[{{{-r, pq[[2]]}, {r, pq[[2]]}},
        {{pq[[1]], -r}, {pq[[1]], r}}}]}]}, Center],
 "Translation of the graph of x^2 by the vector {p,q}",
 {{pq, {0, 0}}, {-r, -r}, {r, r}, Locator},
 Delimiter, {{r, 1, "Range"}, 1, 10}]
```



◻ *Change the function*

```
Manipulate[
 Column[{
   {expr, (expr /. {x -> (x - pq[[1]])}) + pq[[2]]},
   Plot[Evaluate[{expr, (expr /. {x -> (x - pq[[1]])}) + pq[[2]]}],
    {x, -r, r}, AspectRatio → Automatic,
    PlotRange → {{-r, r}, {-r, r}},
    Prolog → {Opacity[0.5], Arrow[{{0, 0}, pq}],
      Line[{{{-r, pq[[2]]}, {r, pq[[2]]}},
        {{pq[[1]], -r}, {pq[[1]], r}}}]}]
  }, Center
 ],
 "Translation of the graph of f(x) by the vector {p,q}",
 {{expr, x^2, "f(x)"}, {x^2, x^3, Sin[x], Cos[x]}},
 {{pq, {0, 0}}, {-r, -r}, {r, r}, Locator},
 Delimiter, {{r, 1, "Range"}, 1, 10}]
```



◻ A quite general version with InputField

```
Manipulate[
 Plot[Evaluate[{expr, (expr /. {x -> (x - p[[1]])}) + p[[2]]}],
  {x, -r, r}, AspectRatio → Automatic,
  PlotRange → {{-r, r}, {-r, r}}, PlotLabel →
   TraditionalForm[(expr /. {x -> (x - p[[1]])}) + p[[2]]],
  Prolog → {Opacity[0.5], Arrow[{{0, 0}, p}],
    Line[{{{-r, p[[2]]}, {r, p[[2]]}},
      {{p[[1]], -r}, {p[[1]], r}}}]}],
 "f(x-p)+q: translation of the graph by the vector {p,q}",
 Delimiter, {{expr, x^2, "f(x)"}, InputField},
 {{r, 1, "Range"}, 1, 10},
 {{p, {0, 0}, "shifth vector"}, {-r, -r}, {r, r}, Slider2D}]
```

f(x−p)+q: translation of the graph by the vector {p,q}

f(x) x^2

Range

shifth vector



Another variation

```
Manipulate[
 Plot[Evaluate[{expr, (expr /. {x -> (x - p[[1]])}) + p[[2]]}],
  {x, -r, r}, AspectRatio → Automatic,
  PlotRange → {{-r, r}, {-r, r}}, PlotLabel →
   TraditionalForm[(expr /. {x -> (x - p[[1]])}) + p[[2]]],
  Prolog → {Opacity[0.5], Arrow[{{0, 0}, p}], Line[{{{-r, p[[2]]},
       {r, p[[2]]}}, {{p[[1]], -r}, {p[[1]], r}}}]}],
 "f(x-p)+q: translation of the graph by the vector {p,q}",
 Delimiter, {{expr, x^2, "f(x)"}, InputField},
 {{r, 1, "Range"}, 1, 10},
 {{p, {0, 0}, "shifth vector"}, {-r, -r}, {r, r}, Locator}]
```

f(x−p)+q: translation of the graph by the vector {p,q}

f(x) x^2

Range



Now, develop something together

Choose a topic:

- *Elementary transformations*
- *Definition or properties of functions*
- *Oscillations, superposition of oscillations*
- *Derivative*
- *Partial derivatives*
- *Anything of common interest*
-

◄ | ►

Before doing anything :

- ***Question: Decide what you want***
  - *Teaching material*
  - *Research material*
  - *Programming exercise*
  - *Something for fun*

◀ | ▶

Question: What is the target group?

- *Yourself*
- *Researchers*
- *Teachers*
- *Students*
- *Anybody*

More questions:

- What is the target professional group (Math, Biology....)?
- Are the users assumed to have knowledge in typesetting, topics, examples considered?

◄ | ►

Purpose 1: Type of material

- *preliminary illustrations*
- *illustrations for the understanding-learning phase*
- *explorations to have deeper knowledge*

Purpose 2: Way of usage

- *classroom illustrations*
- *individual directed study*
- *individual explorations*

◄ | ►

Some Principles

- *Do not forget the target professional group!*
- *Pay attention to the levels of users:*
  - Student's level: What is the main point, what parameters and how should be modified, what are the data, what is the conjecture.....
  - Teacher's (instructor's) level: deeper mathematical and didactic relations, possibilities for explorations, hidden features...
  - Developer's level: nice, well-designed program to help work of other developers.
- *Remember the general principles for choosing the best media*
- *Find the most expressive examples*
- *Design the application according to the main point?*
  - Design the scene and the "story"
  - Fit the best interface, controllers to the problem and the target group
- *Give enough information, hints*
- *Give instructions for interactive study*

Technical tools : Some typical controllers and configurations

Control objects

1D continuous change: Manipulator, Slider, VerticalSlider,Trigger

```
Manipulate[x, {x, 0, 1, Manipulator, Appearance → "Labeled"}]
```




```
Manipulate[x, {x, 0, 1, Slider, Appearance → "Labeled"}]
```




```
Manipulate[x, {{x, 0}, 0, 1, Slider, Appearance → "Vertical"}]
```



Animation

```
Manipulate[x, {x, 0, 1, Manipulator, Appearance → "Labeled"}]
```




```
Manipulate[x, {x, 0, 1, Trigger}]
```




```
Manipulate[Plot[Sin[c x], {x, -Pi, Pi}], {c, 1, 2, Trigger}]
```



```
Manipulate[Plot[Sin[c x], {x, -Pi, Pi}], {c, 1, 2, Animator}]
```



□ *1D discrete changes*

```
Manipulate[x, {x, Opener}]
```




```
Manipulate[x, {x, Checkbox}]
```




```
Manipulate[x, {x, {1, 2}, Checkbox}]
```



```
Manipulate[z, {{z, {1, 2}}, {1, 2, 3, 4}, CheckboxBar}]
```




```
Manipulate[x, {x, RadioButton}]
```




```
Manipulate[x, {x, {1, 2, 3, 4}, RadioButtonBar}]
```




```
Manipulate[x, {x, {1, 2, 3, 4}, SetterBar}]
```




```
Manipulate[x, {{x, {}}, {1, 2, 3, 4}, TogglerBar}]
```



▫ 2D controllers

▫ Slider2D

```
Manipulate[p, {p, Slider2D}]
```



▫ Locator

```
Manipulate[Column[{p,
   Graphics[Point[p], Frame → True, PlotRange → {{0, 1}, {0, 1}}]
  }, Center], {p, {0, 0}, {1, 1}, ControlType -> Locator}]
```



```
Manipulate[Column[{MatrixForm[p],
   Graphics[Point[p], Frame → True, PlotRange → {{0, 1}, {0, 1}}]
  }, Center],
 {{p, {{0, 0}}}, {0, 0}, {1, 1}, Locator, LocatorAutoCreate → True}]
```




```
Manipulate[Column[{MatrixForm[p],
   Graphics[Point[p], Frame → True, PlotRange → {{0, 1}, {0, 1}}]
  }, Center],
 {{p, {{0, 0}}}, {0, 0}, {1, 1}, Locator, LocatorAutoCreate → True},
 Button["Reset", p = {{0, 0}}]]
```



◄ | ►

Some other controllers

Giving expressions: InputField, PopupMenu

```
Manipulate[expr, {{expr, x^2}, InputField}]
```




```
Manipulate[Plot[expr, {x, -1, 1}], {{expr, x^2}, InputField}]
```



Problem : expressions containing parameter

Wrong versions

```
Manipulate[Plot[expr, {x, -1, 1}, PlotLabel → c],
 {{expr, c x^2}, InputField}, {c, 1, 4}]
```



```
Manipulate[Dynamic[Plot[expr, {x, -1, 1}, PlotLabel → c]],
 {{expr, c x^2}, {Sin[c x], Cos[c x], c x^2}, PopupMenu}, {c, 1, 4}]
```



Good versions

```
Manipulate[Manipulate[
  Dynamic[Plot[expr /. c → cc, {x, -1, 1}, PlotLabel → c]],
  {{cc, 1, "c"}, 1, 4}], {{expr, c x^2},
  {Sin[c x], Cos[c x], c x^2}, PopupMenu}]
```



```
Manipulate[Manipulate[
  Dynamic[Plot[expr /. c → cc, {x, -1, 1}, PlotLabel → c]],
  {{cc, 1, "c"}, 1, 4}], {{expr, c x^2},
  {Sin[c x], Cos[c x], c x^2}, SetterBar}]
```



```
Manipulate[Manipulate[
  Dynamic[Plot[expr /. c → cc, {x, -1, 1}, PlotLabel → c]],
  {{cc, 1, "c"}, 1, 4}], {{expr, c x^2},
  {Sin[c x], Cos[c x], c x^2}, InputField}]
```



ColorSetter

```
Manipulate[Plot[expr, {x, -1, 1}, PlotStyle → {colorset}],
 {{expr, x^2}, InputField}, {{colorset, Blue}, ColorSetter}]
```



ColorSlider

```
Manipulate[Plot[expr, {x, -1, 1}, PlotStyle → {colorset}],
 {{expr, x^2}, InputField}, {{colorset, Blue}, ColorSlider}]
```


expr
Sin[x]
colorset
0.5
-0.5
-1.0
-0.5
0.5
1.0

Some special options of Manipulate

```
SaveDefinitions → True (False)
Initialization :→ ()
TrackedsSymbols → {symbols}
ContinuousAction → True (False)
Deployed → True (False)
```